# Práctica 4

# Movimiento Armónico

## 4.1. Enunciado de la Práctica

Ver [AKLo], vol. I, p. 377.

Una partícula de masa 0.5 $[\![\mathrm{kg}]\!]$ se mueve sobre el eje horizontal $Ox$ en un medio resistente o viscoso (por ejemplo, un líquido o un gas) bajo la influencia de la fuerza elástica de dos resortes (ver figura 4.1) que actúan según la ley de Hooke; esta ley establece que la fuerza elástica está dirigida hacia la posición de equilibrio y es proporcional a la desviación a partir de dicha posición de equilibrio. Supongamos que la posición de equilibrio equivale a $x = 0$ y que la fuerza elástica es $-kx\,[\![\mathrm{N}]\!]$ $(k > 0)$.

Figura 4.1: Partícula bajo la acción de la fuerza elástica de dos resortes.

Además, la resistencia del medio es proporcional a la velocidad de la partícula e igual a $-\lambda \dfrac{dx}{dt}\,[\![\mathrm{N}]\!]$ $(\lambda > 0)$.

Sabiendo que la posición inicial de la partícula es la de reposo en $x = 4$, se pide determinar:

a) La ecuación diferencial que describe el movimiento cuando la partícula está sometida, además, a la acción de una fuerza externa de magnitud $E(t)\,[\![\mathrm{N}]\!]$.

b) La posición $x(t)$, $t$ en segundos, de la partícula en el caso $\lambda = 0$, $k = 2, E(t) = 0$.

c) La posición $x(t)$ de la partícula en el caso $\lambda = 0.5$, $k = 2.125, E(t) = 0$.

d) La posición $x(t)$ de la partícula en el caso $\lambda = 0, k = 9, E(t) = 40 \cdot \cos(5t)$.

e) La posición $x(t)$ de la partícula en el caso $\lambda = 0, k = 9, E(t) = 2\cos 2t$.

f) Para $t \in [0, 10]$, calcular la máxima amplitud del movimiento en el caso d).

g) ¿En qué instantes alcanza la partícula la posición de equilibrio en los casos b) y c)?

## 4.2. Ecuación del movimiento

Teniendo en cuenta que la fuerza a la que se ve sometida nuestra partícula coincide con el producto de su masa por su aceleración, i.e., $F = m \cdot \frac{d^2x}{dt^2}$, resulta que:

$$m \cdot \frac{d^2x}{dt^2} = -kx - \lambda \frac{dx}{dt} \tag{4.1}$$

de lo que resulta que la ecuación diferencial que describe el desplazamiento, $x(t)$, de nuestra partícula en función del tiempo $t$ tiene la forma

$$x'' + \frac{\lambda}{m}x' + \frac{k}{m}x = 0, \qquad \lambda \geq 0, k \geq 0 \tag{4.2}$$

Esta ecuación se transforma en la más general

$$x'' + \frac{\lambda}{m}x' + \frac{k}{m}x = \frac{E(t)}{m}, \qquad \lambda \geq 0, k \geq 0 \tag{4.3}$$

en aquellos casos en que además nuestra partícula se encuentra sometida a una fuerza exterior de $E(t)$ $[\![\mathrm{N}]\!]$.

## 4.3. Elementos necesarios

```
> restart;
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

## 4.4. Apartado b: Movimiento armónico simple

```
> ecB:=(D@@2)(x)(t)+4*x(t)=0;
> ini:=x(0)=4,D(x)(0)=0; dsolve({ecB,ini},x(t)); xB:=rhs(%);
> plot(xB,t=0..20);
```

$$ecB := (\mathrm{D}^{(2)})(x)(t) + 4\,\mathrm{x}(t) = 0$$

$$ini := \mathrm{x}(0) = 4,\ \mathrm{D}(x)(0) = 0$$

$$\mathrm{x}(t) = 4\cos(2\,t)$$

$$xB := 4\cos(2\,t)$$




Figura 4.2: Movimiento armónico simple

## 4.5. Apartado c: Movimiento armónico amortiguado

```
> ecC:=(D@@2)(x)(t)+D(x)(t)+(425/100)*x(t)=0;
> dsolve({ecC,ini},x(t)): xC:=rhs(%);
> plot(xC,t=0..20);
```

$$ecC := (\mathrm{D}^{(2)})(x)(t) + \mathrm{D}(x)(t) + \frac{17}{4}\,\mathrm{x}(t) = 0$$

$$xC := e^{(-\frac{t}{2})}\sin(2\,t) + 4\,e^{(-\frac{t}{2})}\cos(2\,t)$$




Figuras 4.3: Movimiento armónico amortiguado

## 4.6. Apartados d,f: Movimiento armónico forzado

```
> ecD:=(D@@2)(x)(t)+18*x(t)=80*cos(5*t);
> dsolve({ecD,ini},x(t)): xD:=rhs(%);
> plot(xD,t=0..20);
```

$$ecD := (\mathrm{D}^{(2)})(x)(t) + 18\,\mathrm{x}(t) = 80\cos(5\,t)$$

$$xD := \frac{108}{7}\cos(3\sqrt{2}\,t) - \frac{80}{7}\cos(5\,t)$$


Figura 4.4: Movimiento armónico con interferencia

Vamos ahora a comprobar que la máxima amplitud del caso d) cuando $t \in [0, 10]$ es $A_m = 26.71686747\,[\![\mathrm{m}]\!]$ para ello utilizamos el listado:

```
> plot(abs(xD),t=0..10);
```




```
> dxD:=diff(xD,t); t0:=fsolve(dxD,{t=4.3});
> Am:=subs(t0,xD);
> Am:=evalf(Am);
```

$$dxD := -\frac{324}{7}\sin(3\sqrt{2}\,t)\sqrt{2} + \frac{400}{7}\sin(5\,t)$$

$$t0 := \{t = 4.420244699\}$$

$$Am := \frac{108}{7}\cos(13.26073410\sqrt{2}) - \frac{80}{7}\cos(22.10122350)$$

$$Am := 26.71686747$$

De la representación gráfica de $|x_D(t)|$, $t \in [0, 10]$, vemos que el máximo se consigue cerca de $t = 4.3$, la exactitud la conseguimos con la instrucción `t0:=fsolve(dxD,{t=4.3});` y el cálculo de la amplitud máxima con las dos últimas instrucciones.

## 4.7. Apartado e: Movimiento armónico en resonancia

```
> ecE:=(D@@2)(x)(t)+4*x(t)=4*cos(2*t);
> dsolve({ecE,ini},x(t)): xE:=rhs(%);
> plot(xE,t=0..20);
```

$$ecE := (\mathrm{D}^{(2)})(x)(t) + 4\,\mathrm{x}(t) = 4\cos(2\,t)$$

$$xE := 4\cos(2\,t) + \sin(2\,t)\,t$$




Figura 4.5: Movimiento armónico en resonancia

Según podemos observar por el gráfico, y con mayor certeza a partir de la expresión de $x_E(t)$, la amplitud de $x_E(t)$ aumenta según lo hace $t$ y, por ello, su amplitud máxima es infinita.

## 4.8. Solución al apartado g

Allí donde la $x$ se hace cero. Tal como hemos visto en el anterior apartado se verifica que las soluciones $x_B(t)$, $x_C(t)$ de los apartados b) y c) tienen, respectivamente, la expresión

$$x_B(t) = 4\cos 2t; \qquad x_C(t) = e^{-t/2}(4\cos 2t + \operatorname{sen} 2t) \tag{4.4}$$

teniendo en cuenta la igualdad $\operatorname{sen}(\alpha + 2t) = \operatorname{sen}\alpha\cos(2t) + \cos\alpha\operatorname{sen}(2t)$ podemos poner

$$\begin{aligned} 4\cos 2t + \operatorname{sen} 2t &= \sqrt{17}(\tfrac{4}{\sqrt{17}}\cos 2t + \tfrac{1}{\sqrt{17}}\operatorname{sen} 2t) \\ &= \sqrt{17}\operatorname{sen}(\alpha + 2t) \end{aligned} \tag{4.5}$$

a condición de que $\alpha = \operatorname{arctg}(4) = 1.3258$, con lo cual

$$x_B(t) = 4\cos 2t; x_C(t) = \sqrt{17}e^{-t/2}(\operatorname{sen}(\alpha + 2t)) \tag{4.6}$$

y ahora ya es fácil observar que las posiciones de equilibrio se consiguen, respectivamente, en los puntos

$$\begin{aligned} P_B &= \{\tfrac{\pi}{4} + k\tfrac{\pi}{2};\quad k \geq 0\} = \{0.7854 + k\tfrac{\pi}{2};\quad k \geq 0\} \\ P_C &= \{\tfrac{k\pi - 1.3258}{2};\quad k \geq 1\} = \{0.9079 + k\tfrac{\pi}{2};\quad k \geq 0\} \end{aligned} \tag{4.7}$$

## 4.9. Animaciones

Damos como ejemplo las instrucciones necesarias para producir la animación del movimiento de la partícula sometida a las condiciones del caso b):

```
> XA:=unapply(xA,t);
```

$$XA := t \to 4\cos(2\,t)$$

```
> for i from 0 to 200 do
> p[i]:=plot([[XA(0.1*i),0]],x=-4.5..4.5,style=point,symbol=circle)
> end do:
> display([seq(p[i],i=0..200)],insequence=true);
```

## 4.10. Espacio de Fases y de Flujo

La ecuación diferencial $ecB$: $\mathrm{D}^2x + 4x = 0$ se puede poner en su forma equivalente:

$$ECb\colon \quad \left\{ \begin{array}{lcl} \mathrm{D}x &=& v \\ \mathrm{D}v &=& -4x \end{array} \right. \tag{4.8}$$

Ahora la representación del campo vectorial $X = v\frac{\partial}{\partial x} - 4x\frac{\partial}{\partial v} \equiv (v, -4x)$ y la pintura de la solución de ecB y de otras dos equivalentes en el espacio de fases (X,V) nos queda:

```
> with(DEtools):

> ECb:=[D(x)(t)=v(t),
> D(v)(t)=-4*x(t)];
```

```
> DEplot(ECb,[x(t),v(t)],
> t=0..20,x=-5..5,v=-10..10, [[x(0)=4,v(0)=0],[x(0)=3,v(0)=0],
> [x(0)=2,v(0)=0]]);
```

$$ECb := [\mathrm{D}(x)(t) = \mathrm{v}(t),\, \mathrm{D}(v)(t) = -4\,\mathrm{x}(t)]$$




Haciendo lo análogo en el caso ecC nos queda:

```
> ECc:=[D(x)(t)=v(t),
> D(v)(t)=-D(x)(t)-(425/100)*x(t)]; DEplot(ECc,[x(t),v(t)],t=0..20,
> x=-5..5,v=-6..6,[[x(0)=4,v(0)=0],
> [x(0)=3,v(0)=0],[x(0)=2,v(0)=0]]);
```

$$ECc := [\mathrm{D}(x)(t) = \mathrm{v}(t),\, \mathrm{D}(v)(t) = -\mathrm{D}(x)(t) - \frac{17}{4}\,\mathrm{x}(t)]$$